```
 1: /**
 2:  * A basic physics simulation using a class to define a
 3:  * Ball object that is under the influence of gravity, and
 4:  * that can collide with other objects.
 5:  *
 6:  * This code is also used as a demonstration of proper coding
 7:  * style.  It shows how to comment the program, variables,
 8:  * classes, and functions.  Additionally, it showcases good
 9:  * variable names, indentation, and whitespace.
10:  *
11:  * @author Jason Healy
12:  */
13:
14: /** A collection of all the balls in the simulation */
15: Ball[] balls = new Ball[10];
16:
17: /** Processing setup function: runs only when the program begins */
18: void setup() {
19:   size(800, 600);
20:   // populate each space in the array with a randomly-generated
21:   // ball
22:   for (int i = 0; i < balls.length; i=i+1) {
23:     balls[i] = new Ball(random(width), random(height));
24:   }
25: }
26:
27: /** Processing draw function: run automatically for each frame */
28: void draw() {
29:   background(0);
30:   // have each ball render itself to the screen
31:   for (int i = 0; i < balls.length; i=i+1) {
32:     balls[i].render();
33:   }
34: }
35:
36: /** Processing mouse event handler
37:  * When the user presses the mouse, impart some additional
38:  * velocity to each of the balls so they jump up on the screen
39:  */
40: void mousePressed() {
41:   background(0);
42:   for (int i = 0; i < balls.length; i=i+1) {
43:     balls[i].pop();
44:   }
45: }
46:
47:
48: /**
49:  * A Ball is represented by a circle on the screen.  Every Ball
50:  * moves with its own velocity, and is capable of being influenced
51:  * by gravity and by colliding with the sides of the screen and
52:  * other Balls.
53: */
54: class Ball {
55:   /** Radius of each ball */
56:   int radius = 25;
57:
58:   /** Array of possible colors (alternated when it collides) */
59:   color[] c = {
60:     color(255, 0, 0, 255), color(0, 255, 0, 255), color(0, 0, 255, 255)
61:   };
62:
63:   /** X location of the Ball on the screen */
64:   private float x;
65:   /** Y location of the Ball on the screen */
```

```
66:   private float y;
67:
68:   /** X velocity */
69:   private float vx;
70:   /** Y velocity */
71:   private float vy;
72:
73:   /** Timestamp (millis) since the last update; used to
74:    *  calculate accelerations and velocities */
75:   private int last;
76:
77:   /** Number to track which color to use for the Ball;
78:    *  it is an index into the colors array "c" */
79:   private int colour = 0;
80:
81:   /** Tracks whether this Ball has been pressed by the mouse */
82:   private boolean pressed = false;
83:
84:
85:   /** Constructor.  Initializes Ball at the given coordinates
86:    *  and assigns a random initial velocity. */
87:   Ball(float x, float y) {
88:     this.x = x;
89:     this.y = y;
90:     vx = random(-width/10, width/10)*10;
91:     vy = random(-height/10, height/10)*10;
92:     last = millis();
93:   }
94:
95:   /** Updates the Ball based on physics engine, and then
96:    *  renders the Ball to the screen. */
97:   void render() {
98:     // determine how long it's been since the last update
99:     int now = millis();
100:    float elapsed = (now - last) / 1000.0;
101:    last = now;
102:
103:    // change position based on velocity and elapsed time
104:    x += elapsed * vx;
105:    y += elapsed * vy;
106:
107:    // change y velocity based on gravity
108:    vy += elapsed * 1000;
109:
110:    // bounce off the left and right sides of the screen
111:    if ( (x < radius && vx < 0) || (x > width-radius && vx > 0) ) {
112:      vx *= -0.9;
113:    }
114:    // bounce off the top and bottom sides of the screen
115:    if ( (y < radius && vy < 0) || (y > height-radius && vy > 0) ) {
116:      vy *= -0.9;
117:    }
118:
119:    // try to collide with every single other Ball in the program
120:
121:    // start with the first Ball in the array
122:    int i = 0;
123:    // loop through all the Balls "before" this one in the array
124:    // and ignore them
125:    while (balls[i] != this) {
126:      i=i+1;
127:    }
128:    // now skip "this" Ball (so we don't collide with ourself)
129:    i=i+1;
130:
```

```
131:     // finally, actually try to collide with all the remaining
132:     // Balls in the array.  Because we skipped the ones "before"
133:     // us in the array (and all the others will do the same),
134:     // only one collision between any two Ball objects is ever
135:     // attempted.
136:     while (i < balls.length) {
137:       collide(balls[i]);
138:       i=i+1;
139:     }
140:
141:     // now that all the physics updates are done, render the
142:     // ball on screen
143:     noStroke();
144:     fill(c[colour]);
145:     ellipse(x, y, radius*2, radius*2);
146:   }
147:
148:   /** Rotate to the next color in the array. */
149:   void bump() {
150:     colour = (colour+1)%c.length;
151:   }
152:
153:   /** Give a random "kick" in y velocity to this Ball. */
154:   void pop() {
155:     vy = random(-height, 0)*2;
156:   }
157:
158:   /** Attempt to collide with another Ball object (passed
159:    *  as a parameter). */
160:   void collide(Ball o) {
161:     // see how far apart the centers of both Ball objects are
162:     float overlap = 2*radius - dist(x, y, o.x, o.y);
163:
164:     if (overlap < 0) {
165:       // no collision if they aren't touching...
166:       // just return without doing anything else
167:       return;
168:     }
169:
170:     // If we didn't return above, there must be some overlap
171:
172:     // get the angle that this ball collides with the other
173:     float theta = atan2(o.y-y, o.x-x);
174:
175:     // generate a force (velocity change) along the axis
176:     // that connects the two balls, pushing away from the
177:     // point of contact
178:     // 7-10 is a fudge factor to make the bounces look good
179:     vx -= cos(theta) * radius * 10;
180:     vy -= sin(theta) * radius * 10;
181:     o.vx += cos(theta) * radius * 10;
182:     o.vy += sin(theta) * radius * 10;
183:
184:     // move the balls apart so they're no longer touching
185:     x -= cos(theta) * overlap / 2;
186:     y -= sin(theta) * overlap / 2;
187:     o.x += cos(theta) * overlap / 2;
188:     o.y += sin(theta) * overlap / 2;
189:
190:     // tell each Ball to "bump" so its color changes
191:     bump();
192:     o.bump();
193:   }
194: }
```